Interfacing Laser and Computer to Monitor Animal Behaviors

By John Scott Christianson and Jim Rose

This document consists of the abstracts, methodologies, results, and discussions of two experiments in monitoring animal behavior with a computer and helium-neon laser. The first experiment deals with monitoring animals in two dimensions and the second experiment in three dimensions. This document is public domain and may not be used for commercial gain. This document may be distributed freely as long as the entire document is distributed without any changes.

Our purpose is to provide a reference for other ideas and experiments. We would like to thank Mr. Tom Eddy and Mr. Frank Kiener, two exceptional teachers who gave us the support to persue our projects. We would especially like to thank our parents for their support and encouragement. John O. Christianson also put in many hours making the box container for the 3- Dimensional monitoring device. We would also like to thank James E. Rathke, Assistant Professor, at the University of Missouri-Columbia for designing the optocoupler.

This research was presented to the 1986 and 1987 Wisconsin Science Congress. We would appreciate knowing about any experiments or ideas this leads others to do, as we hope that this document will help. If so please drop us note. Questions or comments may be directed to:

John Scott Christianson
709 Green Meadows
Columbia, MO 65203
Bitnet: C487629@UMCVMB

OR

Jim Rose
R. R. # 2 Box 170
Ripon, Wi 54971

Jim Rose and John Scott Christianson

Table of Contents

Abbreiations used in text.

DPDT: Double Pole Double Throw
I/O Connector: Input and Output Connector
LED: Light Emiting Diode
I-R Light: Infared Light (Wave length of 0.7 micrometre to 1 millimetre)
SLR: Single Lens Reflex

Monitoring Animals in Two Dimensions

Abstract

The objective of this investigation is to construct a device that can monitor the movement/behaviors of small confined animals. We hypothesized that a laser interfaced with an Apple IIe™ can be used to monitor the movement of a small animal. Behavioral responses by an animal to controlled environmental variables can be monitored and recorded by the experimenter using such a device. Moreover, the experimenter need not be present when monitoring/recording is taking place.

The computer is able to detect light through a phototransistor connected to the game port. The Apple IIe™ game port has 16 pins, 11 that can be indexed with seven that can be used for input.

We constructed a grid made of laser beams one-half inch apart to form a three inch by three inch area. This area consists of six beams on the X-axis and six beams on the Y—axis. When the laser "hits" an I-R phototransistor, the computer determines that the beam is not interrupted and thus, the animal is not in that particular grid area.

The computer then indexes the I-R phototransistors on the X-axis and determines which are interrupted. The computer stores this data, indexes separate relays to switch input to the Y-axis, and runs the program to determine which beams are interrupted on the Y-axis. From this information the computer can determine where the animal is located; plot the location on the computer screen; and at programmed intervals store the data in memory to be printed out for later analysis.

Methodology

All of the data from the I-R phototransistors enters the computer through the I/O game port connector, see Appendix 2, Figure-1. Data can be received through 7 of the pins. Pins 2, 3, and 4 (PB0, PB1, PB2), of the game port, are single bit inputs and are used to detect on or off conditions. I-R photodetectors were connected to the PB0 and PB1 pins. The I-R photodetectors are switches. TIL 414 I-R phototransistors were connected through GC0, GC1, GC2, and GC3. The computer reads a resistance value through these pins. The TIL 414 I-R phototransistors vary in resistance according to the intensity of the light hitting them.

Because the computer could only receive information through 7 of the 16 pins, we used two rows each consisting of four I-R phototransistors and two I-R photodetectors to monitor the X and Y axis. We used a relay to switch between the X and Y axis inputs, thereby using the same 6 inputs to monitor both X and Y axis. The relay was controlled through an optocoupler (Appendix 2, figure 2). A 6V lattern battery powered the relays. The LED in the optocoupler is turned on and off through the AN3 pin. This is done in the software by POKEing different values into memory addresses (see Appendix I). An annunciator pin's output is 3.5V when activated. The optocoupler was designed by James E. Rathke, Associate Professor, Circuits, Electrical and Computer Engineering Dept., at the University of Missouri-Columbia.

An interface box was constructed (Appendix III, picture 1). The box housed the optocoupler, 3 DPDT relays, and RCA jacks to connect to the rows of I-R photodetectors/transistors. The schematic for the box is in Appendix II (Figure-3), as are the specs on the relays used. The interface box is connected to the Game Port I/O connector, in the computer, through a 16 wire ribbon cable.

The RCA jacks allowed for easy trouble shooting and allows the box to be used for other experiments. The arrangement of I-R photodetectors/transistors in relation to the X and Y axis is diagramed in Figure 4 of Appendix II. It is extremely important to maintian the relationship as diagramed because the computer is "expecting" the I-R photodetectors/transistors to be in the same arrangement.

We used a Class c, helium-neon laser. The laser is aimed at the main splitter which splits the beam into two beams at a 90 degree angle from one another. We used microscope slides for beam spitters. Both beams then enter through a series of 6 splitters to form the grid. A frame was constructed to hold the rows of photodetectors/transistors and the beam splitters. Notches, bigger than the splitters were cut in the frame and were filled with modeling clay. The splitters were then placed in the clay and adjusted so that the beams hit the photodetectors/transistors. This produces 36 different points at which the organism can be found. This offers the experimenter more data than the common device used, the tilt board. The tilt board can only register 8 places that an organism can be in. In the tilt board when an organism moves the board tilts due to the organisms weight and triggers one of 8 switches. Photography is also used to monitor organisms by using a slow speed movie camera, or a SLR, with an autowind, on a peroidic timer.

After using program #1, ALINE TEST, to make sure that the laser beams are correctly alined the computer starts collecting data. The computer writes the data to disk as it is collected. The data can be reviewed later as a graphic representation of the area that was monitored. The data can be reviewed in real time (same amount of time as collected) or speeded up.

With alterations to the viewing program, a search can be initiated for certain movements or patterns. The graphics can also be sent to a dot matrix printer for hard copy. The computer can record data indefinitely, with the only limiting factor being disk space. The programs in Appendix I are for use with a two 5 1/4 drive system. Different system setup is described in Appendix I.

Results and Discussion

Our setup worked well. We didn't use this device on an animal, due to concern of blinding the animal with the laser. We tested the device by using our fingers to interrupt the beams.

The alinement of the laser and I-R photodetectors was, at times, difficult. The laser had to be hitting the I-R photodetectors exactly, in order for the I-R light to trigger the I-R photodetectors to an "on" condition. This was mainly due to the fact that one laser was powering 12 beams.

Damage to the organism's retina is now our main concern. I-R light has been shown to damage the retina. The critical point in avoiding damage would seem to be placing the grid at a level that the eyes will not be at, i.e., place the laser grid at the lowest part of the cage. The use of different photodetectors/transistors that would work on a non-harmful wave length is also a possibility.

The computer programs worked very well. The next step in the programming would be a program that would detect patterns in the movement of the organism. If the grid is placed at the bottom of the cage, data points might be missing, due to the higher chance that the feet of the animal will not interrupt a beam on the grid. A computer program that could trace where an animal was and where it next showed up at, and could determine the probable path in between, would be useful in future applications.

Compiling the BASIC code into machine language would improve the programs speed.

With some of the technology from our attempt at monitoring animals in three dimentions (Calling Four™ game port expanders) the two dimentional grid and data points could be expanded, to monitor a greater area with the same presision.

We believe that our device, as constructed, serves the purpose of: 1). collecting data about the position, x and y, of an organism; 2). being able to function without human supervision; 3). providing more information to the experimenter, then previously used methods (tilt board and photography)

We can see several practical applications of this device. This design and variations on this design is useful when recording any Ethological (behavior) responses of an organism in a controlled environmental experiment. The advantages might be better seen when dealing with: 1). experiments with radiation, as the computer and laser would not be affected as would camera film would; 2) monitoring aquatic organisms i.e. monitoring the sensing ability of a blind cave fish, by putting a "bubble net" in the middle of the animal's tank.

Appendix I: The Computer Programs

All of these programs were written using DOS 3.3. Adaptation to ProDOS is possible by changing the read/write operations to fit the syntax of that operating system.

The configuration of the Apple IIe ™ system that was used is as follows: Apple DMP printer with parallel card in slot #1, two disk drives in slot #6, and a monochrome monitor. The program disk is put into drive #1, and an initialized disk is put in drive #2. The file DATA TD will be written to the disk in drive #2.

A different disk drive configuration can be accommodated by replacing D2 with D1, to write the data to the first drive. This will, of course, mean a reduction in the amount of data that can be stored, if D1 also contains the rest of the programs.

Pins 2, 3, and 4 are known as single bit inputs, as said before, they detect "ON" or "OFF" situations. The computer can tell if a switch connected to one of these pins is on or off. If the switch is on, a value greater than 120 will be given at -16287 for pin 2, -16286 for pin 3, and -162185 for pin 4, i.e., 100 IF PEEK (-16287) > 120 THEN PRINT "Beam not interrupted." In these programs an "ON" condition indicates that the laser beam is not broken.

Pins 6, 10, 7, and 11 monitor resistance. "The computer reads a number (0 to 255 or 8-bit resolution) which is proportional to the resistor (about 200 to 120,000 ohms for the full 0 to 255 range) attached between pin 1 (+5 volts) and the game paddle pin. The value is best read from BASIC by setting the varible equal to PDL(0) for Pin 6, PDL(1) for Pin 10, PDL(2) for Pin 8, and PDL(3) for Pin11."- A Compendium of Introductory Apple II Interfacing Examples for Science Education Application, by Roy Knispel.

In this experiment the beam was not interrupted if the value of PDL(0-3) is less than 120, i.e., 100 IF (INT(PDL(0))) < 120 THEN PRINT "Beam not interrupted."

Pins 15, 14, 13, and 12 are output switchs or annunciator pins. Low and high values may be set on these pins. A high value is induced by POKEing a value of 1 into -16296 for pin 15, -16294 for pin 14, -16292 for pin 13, and -16290 for pin 12, i.e., 100 POKE -16296,1 : REM Turns pin 15 high.

A high pin has a voltage of 3.5V. A low pin has a voltage less than 0.3 volts. A low value is induced by POKEing a value of 0 into -16295 for pin 15, -16293 for pin 14, -16291 for pin 13, and -16289 for pin 12, i.e., 100 POKE -16295,0 : REM Turns pin 15 low.

These high and low conditions remain until the computer is told to reverse the condition. The Annunciator pins are always low after booting the computer, until instructions are given to make them high.

We are willing to send copies of the following programs to any persons interested. Just send a blank 5 1/4" disk and a Self Addressed Stamped disk mailer (or similar package). The programs DUMP.HIRES and PRINTER.INFO must be obtained from the Beagle Brothers™ Triple-Dump disk.

Description of programs

HELLO: Start up program. allows the user to select which program to run.

ALINE TEST TD: Allows the user to check for the proper alinement of the laser with I-R photodetectors/transistors. Must be run before collecting data.

COLLECT DATA TD: Writes the data to disk. User can select the number of readings and delay time between each reading. Data is written as a sequential text file and can also be viewed with programs such as Applewriter.

PRINT DATA TD: Prints the data to the printer for hard copy.

DISPLAY DATA TD: Presents a graphical representation of the area being monitored.

PRINT DISPLAY TD: Used to print the display to a Dot matrix printer. Screen dumping is from Beagle Brothers™ Triple-Dump disk. To set up for a printer other than an Apple DMP, use Triple-Dump to change the PRINTER.INFO file.

DUMP.HIRES: A support file for PRINT DISPLAYTD

PRINTER.INFO: A support file for PRINT DISPLAY TD

DATA TD: A sequential data file on, Drive 2, disk. Where the data from the experiment is stored.

HELLO

```
5 TEXT: HOME: LET T2$="Menu":LET T2$="______": VTAB 5: HTAB (40-((LEN(T1$))/2)):
PRINT T1$
10 VTAB 4 :HTAB (40-((LEN(T1$))/2)): PRINT T2$: PRINT: PRINT
15 HTAB 3:PRINT "1). Collect Data"
20 HTAB 3:PRINT "2). Print Data"
25 HTAB 3:PRINT "3). Display Data"
30 HTAB 3:PRINT "4). Print Display"
35 HTAB 3:PRINT "99). Quit"
40 VTAB 23: PRINT"Type Number of Program"
45 VTAB 23: HTAB 28:INPUT "";N$: N=VAL (N$): IF N=0 THEN 40
50 IF N=99 THEN 95
55 IF N<1 OR N>4 THEN 40
60 HOME: VTAB 10: HTAB 5 : PRINT "One Moment Please-"; : FLASH : PRINT "Loading": NORMAL
65 D$=CHR$(4)
70 PRINT D$;"RUN";: ON N GOTO 75,80,85,90
75 PRINT "ALINE TEST TD"
80 PRINT "PRINT DATA TD"
85 PRINT "DISPLAY DATA TD"
90 PRINT "PRINT DISPLAY TD"
95 HOME: VTAB 15: PRINT "Thank You !"
100 END
```

ALINE TEST TD

```
5 REM ALINE TEST TD program
10 REM This program checks for the proper alinement of laser and
15 REM phototransistor
20 REM
25 REM February, 1986
30 REM by John Scott Christianson
35 REM
40 REM Switch to X-axis
45 POKE -162990,1
50:
55 HOME
60 PRINT"Aline the laser with the"
65 PRINT"Phototransistors/detectors on the X-axis."
70 PRINT"Press <Return> to begin alinement"
75 PRINT"test.  After X-axis is checked,"
80 PRINT"press any key to check Y-axis."
85 REM
90 :INPUT N$: REM Look for the <Return> key
95 REM
100 REM Set up variables for game port
105 A=PEEK (-16287): FOR PAUSE=1 TO 10: NEXT : REM PB0
110 S=PEEK (-16286): FOR PAUSE=1 TO 10: NEXT :REM PB1
115 R=INT (PDL(0)): FOR PAUSE=1 TO 10: NEXT :REM GC0
120 L=INT (PDL(1)): FOR PAUSE=1 TO 10: NEXT :REM GC1
125 G=INT (PDL(2)): FOR PAUSE=1 TO 10: NEXT :REM GC2
130 F=INT (PDL(3)): FOR PAUSE=1 TO 10: NEXT :REM GC3
135 W$="OFF":V$="OFF":O$="OFF":L$="OFF"
140 J$="OFF":M$="OFF"
145 REM
150 REM Check values to determine if
```

```
155 REM Light is hitting the phototransistor
160 REM
165 IF A >120 THEN W$="ON"
170 IF S >120 THEN V$="ON"
175 IF R < 120 THEN O$="ON"
180 IF L < 120 THEN L$="ON"
185 IF G < 120 THEN J$="ON"
190 IF F < 120 THEN M$="ON"
195 :
200 HOME
205 REM Print Status
210 VTAB 1:PRINT W$ : :VTAB 5: PRINT V$: VTAB 9 : PRINT O$: VTAB 13: PRINT L$: VTAB 17 :
PRINT J$: VTAB 21: PRINT M$
215 REM
220 REM Check to see if user wants to move on
225 :
230 T=PEEK (-16384)
235 POKE -16368,0
240 IN T> 128 THEN 260
245 REM If not then repeat
250 GOTO 105
255 REM Switch to Y-axis
260 POKE -16289,0
265 HOME
270 PRINT"Aline the laser with the"
275 PRINT" Phototransistors/detectors on the Y-axis."
280 PRINT"Press <Return> to begin alinement"
285 PRINT"test.  After Y-axis is checked,"
290 PRINT"press any key to start collecting data."
295 :INPUT N$: REM Look for the <Return> key
300 REM Set up variables for game port
305 A=PEEK (-16287): FOR PAUSE=1 TO 10: NEXT : REM PB0
310 S=PEEK (-16286): FOR PAUSE=1 TO 10: NEXT :REM PB1
315 R=INT (PDL(0)): FOR PAUSE=1 TO 10: NEXT :REM GC0
320 L=INT (PDL(1)): FOR PAUSE=1 TO 10: NEXT :REM GC1
325 G=INT (PDL(2)): FOR PAUSE=1 TO 10: NEXT :REM GC2
330 F=INT (PDL(3)): FOR PAUSE=1 TO 10: NEXT :REM GC3
335 W$="OFF":V$="OFF":O$="OFF":L$="OFF"
340 J$="OFF":M$="OFF"
345 REM
350 REM Check values to determine if
355 REM Light is hitting the phototransistor
360 REM
365 IF A >120 THEN W$="ON"
370 IF S >120 THEN V$="ON"
375 IF R < 120 THEN O$="ON"
380 IF L < 120 THEN L$="ON"
385 IF G < 120 THEN J$="ON"
390 IF F < 120 THEN M$="ON"
395 :
400 HOME
405 REM Print Status
410 REM COPY LINE 350 FROM BUG TEST
415 REM
420 REM Check to see if user wants to move on
425 :
430 T=PEEK (-16384)
435 POKE -16368,0
440 IN T> 128 THEN 460
```

```
445 REM If not then repeat
450 GOTO 305
```



```
455 REM Run collect data program
460 D$=CHR$(4): PRINT D$ "RUN COLLECT DATA TD": END
```

COLLECT DATA TD

```
5 REM COLLECT DATA TD Program
10 REM This program collects the data and
15 REM writes it to a text file
20 REM
25 REM by John Scott Christianson and Jim Rose
30 REM
35 HOME
40 REM Initialize Variables
45 D$=CHR$(4)
50 P =10: REM Delay between measurements
55 REM
60 PRINT "How many sets of data do"
65 PRINT "you want to collect";:Input b
70 B=B+1
75 PRINT "Input delay between readings"
80 PRINT "0= no delay, the greater the number"
85 PRINT "the greater the delay";:INPUT Q
90 Q=Q+2
95 REM
100 REM Data is stored in the text file
105 REM "DATA TD" on disk drive # 2
110 REM of a two drive system.
115 PRINT D$ "OPEN DATA TD",D2
120:
125 FOR M=1 TO B
130 GOSUB 1000
135 GOSUB 2000
140 FOR U= 1 TO  Q: NEXT Q
145 NEMT M
150 :
155 REM Close the File and End
160 PRINT D$"CLOSE DATA TD",D2
165 PRINT D$ "RUN HELLO": END

1000 REM Get data
1005 REM Switch to X-axis
1010 POKE -16290,1
1015 :
1020 X1=PEEK (-16287):FOR J=1 TO P: NEXT J
1025 X2=PEEK (-16286):FOR J=1 TO P: NEXT J
1030 X3=INT (PDL(0)):FOR J=1 TO P: NEXT J
1035 X4=INT (PDL(1)):FOR J=1 TO P: NEXT J
1040 X5=INT (PDL(2)):FOR J=1 TO P: NEXT J
1045 X6=INT (PDL(3)):FOR J=1 TO P: NEXT J
1050 REM
1055 REM Sitch to Y-axis
1060 Y1=PEEK (-16287):FOR J=1 TO P: NEXT J
1065 Y2=PEEK (-16286):FOR J=1 TO P: NEXT J
```

```
1070 Y3=INT (PDL(0)):FOR J=1 TO P: NEXT J
1075 Y4=INT (PDL(1)):FOR J=1 TO P: NEXT J
1080 Y5=INT (PDL(2)):FOR J=1 TO P: NEXT J
1085 Y6=INT (PDL(3)):FOR J=1 TO P: NEXT J
```



```
1090 REM
1095 REM Return
1100 RETURN

2000 REM Write data to text file
2005 PRINT D$"WRITE DATA DT",D2
2010 PRINT X1,X2,X3,X4,X5,X6
2015 PRINT Y1,Y2,Y3,Y4,Y5,Y6
2020 REM Return
2025 RETURN
```

PRINT DATA TD

```
5 HOME: D$=CHR$(4)
10 PRINT: PRINT "Input Data sets to print"
15 INPUT "Enter Beginning";B1
20 INPUT "Enter End"; B2
25 IF B2>B1 THEN PRINT "END BIGGER THAN BEGINNING !!": GOTO 10
30 PRINT D$"PR#1"
35 PRINT D$ "OPEN DATA TD",D2
40 PRINT D$ "READ DATA TD", D2
45 Q=0
50 Q=Q+1
55 INPUT X1,X2,X3,X4,X5,X6
60 INPUT Y1,Y2,Y3,Y4,Y5,Y6
65 IF Q=B1 THEN 75
70 GOTO 50
75 PRINT X1;TAB (6);X2 TAB(12);X3 TAB(18);X4 TAB(24);X5 TAB(30);X6
80 PRINT Y1;TAB (6);Y2 TAB(12);Y3 TAB(18);Y4 TAB(24);Y5 TAB(30);Y6
85 PRINT
90 INPUT  X1,X2,X3,X4,X5,X6
95 INPUT Y1,Y2,Y3,Y4,Y5,Y6
100 IF Q=B2 THEN 115
105 Q=Q+1
110 GOTO 75
115 PRINT
120 PRINT D$"CLOSE DATA TD",D2
125 PRINT D$
130 PRINT D$ "PR#0"
135 HOME:PRINT
140 PRINT "Want to print some more ";
145 Input P$
150 IF P$="Y" or P$="YES" THEN 10
155 IF P$="N"or P$="NO" THEN 165
160 GOTO 135
165 PRINT D$"RUN HELLO"
```

DISPLAY DATA TD

```
5 HOME
10 REM Initalize Varibles
15 D$=CHR$(4)
20 REM Get sets to display
25 PRINT: PRINT "Input Data Sets to Display"
```



```
30 INPUT "Enter Beginning"; B1
35 INPUT "Enter End"; B2
40 IF B2>B1 THEN PRINT "END BIGGER THAN BEGINNING !!": GOTO 25
42 PRINT "Enter delay time between each display"
43 INPUT "0=No Delay, greater the # the greater the delay";F
45 REM Read and Count until Beginning
50 PRINT D$"OPEN DATA TD",D2
55 PRINT D$"READ DATA TD",D2
60 Q=0
65 Q=Q+1
70 INPUT X1,X2,X3,X4,X5,X6
75 INPUT Y1,Y2,Y3,Y4,Y5,Y6
80 IF Q=B1 THEN 90
85 GOTO 65
90 REM Print on Screen
95 HGR: HCOLOR=6
100 HPLOT 70,150 TO 210,150 TO 210,10 TO 70,10 TO 70,150
105 IF X1<120 THEN X=90: GOSUB 200
110 IF X2<120 THEN X=110: GOSUB 200
115 IF X3>120 THEN X=130: GOSUB 200
120 IF X4>120 THEN X=150: GOSUB 200
125 IF X5>120 THEN X=170: GOSUB 200
130 IF X6>120 THEN X=190: GOSUB 200
135 Q=Q+1
140 IF Q=B2 THEN 400
145 FOR G= 1 TO F : NEXT G
150 GOTO 100

200 IF Y1<120 THEN Y=30: GOSUB 300
205 IF Y2<120 THEN Y=50: GOSUB 300
210 IF Y3>120 THEN Y=70: GOSUB 300
215 IF Y4>120 THEN Y=90: GOSUB 300
220 IF Y5>120 THEN Y=110: GOSUB 300
225 IF Y6>120 THEN Y=130: GOSUB 300
230 RETURN

300 HCOLOR=3: HPLOT X-5,Y-5 TO X+5,Y-5 TO X+5,Y+5 TO X-5, Y+5 TO X-5,Y-5
305 RETURN

400 PRINT D$"CLOSE DATA TD",D2
405 PRINT D$"RUN HELLO"
```

PRINT DISPLAY TD

```
5 HOME
10 REM Initalize Varibles
15 A=25088
20 REM Lines 25-110 and 500-610 are from the Beagle Brothers Triple-Dump Disk
25 PRINT CHR$(4)"BLOAD DUMP.HIRES,A"A
```

```
30 PRINT CHR$(4)"BLOAD PRINTER.INFO,A"A-512
35 S=1
40 P=1
45 T=0
50 B=192
55 L=0
60 W=280
65 I=0
70 SC=0
```



```
75 LF=1
80 TY=0
85 R=0
90 N=1
95 D=0
100 MX=1
105 MY=1
110 DE=0
115 D$=CHR$(4)
120 REM Get sets to display
125 PRINT: PRINT "Input Data Sets to Display"
130 INPUT "Enter Beginning"; B1
135 INPUT "Enter End"; B2
140 IF B2>B1 THEN PRINT "END BIGGER THAN BEGINNING !!": GOTO 125
142 PRINT "Enter delay time between each display"
143 INPUT "0=No Delay, greater the # the greater the delay";F
145 REM Read and Count until Beginning
150 PRINT D$"OPEN DATA TD",D2
155 PRINT D$"READ DATA TD",D2
160 Q=0
165 Q=Q+1
170 INPUT X1,X2,X3,X4,X5,X6
175 INPUT Y1,Y2,Y3,Y4,Y5,Y6
180 IF Q=B1 THEN 190
185 GOTO 165
190 REM Print on Screen
195 HGR: HCOLOR=6
200 HPLOT 70,150 TO 210,150 TO 210,10 TO 70,10 TO 70,150
205 IF X1<120 THEN X=90: GOSUB 260
210 IF X2<120 THEN X=110: GOSUB 260
215 IF X3>120 THEN X=130: GOSUB 260
220 IF X4>120 THEN X=150: GOSUB 260
225 IF X5>120 THEN X=170: GOSUB 260
230 IF X6>120 THEN X=190: GOSUB 260
235 Q=Q+1
240 IF Q=B2 THEN 400
245 FOR G= 1 TO F : NEXT G
250 GOSUB 500
255 GOTO 200

260 IF Y1<120 THEN Y=30: GOSUB 300
265 IF Y2<120 THEN Y=50: GOSUB 300
270 IF Y3>120 THEN Y=70: GOSUB 300
275 IF Y4>120 THEN Y=90: GOSUB 300
280 IF Y5>120 THEN Y=110: GOSUB 300
285 IF Y6>120 THEN Y=130: GOSUB 300
290 RETURN

300 HCOLOR=3: HPLOT X-5,Y-5 TO X+5,Y-5 TO X+5,Y+5 TO X-5, Y+5 TO X-5,Y-5
```

```
305 RETURN

400 PRINT D$"CLOSE DATA TD",D2
405 PRINT D$"RUN HELLO"

500 POKE A-257,LF *128
505 POKE A-1,S
510 POKE A+3,R * 128
515 POKE A+4,N * 128
520 POKE A+5,D
525 POKE A+6,P * 32
```



```
530 POKE A +7,T
535 POKE A +8,B
540 POKE A +9,L- INT (L/256) *256
545 POKE A +10, INT (L/ 256)
550 POKE A +11, W- INT (W/256) * 256
555 POKE A +12, INT (W/256)
560 POKE A +13, MX
565 POKE A +14, MY
570 POKE A +15, I-INT (I/256)
575 POKE A +16, INT (I/256)
580 POKE A +17,SC * 128
585 POKE A +18, DE
590 POKE A +19,TY
595 POKE 49232, 0: POKE 49234,0 : POKE 49239,0
600 POKE 49236 + P -1,0
605 POKE -16289,0
610 CALL A :REM PRINT IT !!
615 TEXT:HOME
620 RETURN
```

Appendix II: Electronic Schmatics

Figure-1



Table-1

Game Port I/O Connector Descriptions		
Pin	Function	Description
1	+5V	+5v, don't excede 100 mA drain
2-4	PB0-PB2	Single-bit inputs,
5	C040 Strobe	A strobe line
6,7,10,11	GC0-GC3	Game controller inputs
8	Gnd	Electrical Ground
12-15	AN0-AN3	Annunciator outputs
9,16	NC	No internal connection

Figure-2

Optocoupler



Parts

R-1: 62K, 1/4 Watt Resistor
R-2: 2.2K, 1/4 Watt Resistor
LED-1: Red LED
PT-1: FPT-100 Phototransistor
T-1: RCA SK3854 or 2N3904, small signal type transistor
T-2: RCA SK3044 or .5A or greater I
D-1: Coil Diode, 1N34A or similar
Ry-1: Relay, Radio Shack 275-240

Table-2 **Data on Optocoupler Parts**

Relay (Micromini) Radio Shack 275-240 Contacts= 1amp at 125 VAC Coil: 5VDC, 90ma, 55 Ω	RCA SK3854--123AP NPN Silicon Transistor to-92 case Pt= 1.2W Ic= .8A Vcbo= 75V Vceo= 40V Vebo= 6v Hfe= 200 Min Ft= 300 MHz
FPT100- Silicon phototransistor Radio Shack 276-130 to-5 case operating juction temp: -55 C to +85 C Max. power dissipation: 100 mW Iceo (collector dark current): typ 20 ma	
RCA SK3044--154 NPN Silicon transistor TV Chroma Amp to-39 case Pt= 10W Ic= 1A Vcbo= 300V Vceo= 300V Vebo= 7V Hfe= 80 Ft= 30 MHz	

2-D Interface Box

Bottom View

Figure-3



Table-3

Interface Box Parts

R-1: 10K ohm resistor
Ry 1-3: 5VDC Relay
Radio Shack 275-215
Coil Resistance: 45 ohms ±10%
Pull in Voltage: 3.75 VDC
Contact Rating: 1 amp at 125 VAC
Pull in Time: < 6mS
Absolute maximum ratings
Ambient Temperature: -25° C to +60° C
Continuous Coil Voltage: 7.5 VDC
Coil Dissipation: 1W at 20° C

Figure-4



Appendix III: Pictures

Picture-1 Interface Box

Picture 2: Frame with Splitters and Photodetectors/transistors